Example 5b: Multivariate Change and Lagged Effects via Multivariate MLM: Multilevel SEM ("MLM" below) and Single-Level SEM in Mplus v. 8.1+ (complete syntax and output available electronically)

These simulated data are from Hoffman (2015) chapter 9 and include 200 girls measured approximately annually from ages 12–18 (time 0 = age 18) on their risky behavior (the outcome, a sum ranging from 10 to 50) and the extent to which their mothers monitored their activities (the time-varying predictor, a mean ranging from 1 to 5, centered at 3). A time-invariant predictor of the conservativeness of mothers' attitudes about the smoking and drinking (a mean ranging from 1 to 5, centered at 4) was also collected at the age 12 occasion. Here are the individual change trajectories for risky behavior and monitoring:







Multivariate Multilevel Model 1

Level 1 :

$$y_{tid} = dvR\left[\beta_{0iR} + \beta_{1iR}\left(Age_{tiR} - 18\right) + \beta_{2iR}\left(Age_{tiR} - 18\right)^2 + e_{tiR}\right] + dvM\left[\beta_{0iM} + \beta_{1iM}\left(Age_{tiM} - 18\right) + e_{tiM}\right]$$

Level 2 :

Risky Intercept: $\beta_{0iR} = \gamma_{00R} + \gamma_{01R}\left(Attitudes12_i - 4\right) + U_{0iR}$

Risky Age: $\beta_{1iR} = \gamma_{10R} + \gamma_{11R}\left(Attitudes12_i - 4\right) + U_{1iR}$

Risky Age^2: $\beta_{2iR} = \gamma_{20R}$

Monitor Intercept: $\beta_{0iM} = \gamma_{00M} + \gamma_{01M}\left(Attitudes12_i - 4\right) + U_{0iM}$

Monitor Age: $\beta_{1iM} = \gamma_{10M} + \gamma_{11M}\left(Attitudes12_i - 4\right) + U_{1iM}$

The best-fitting unconditional longitudinal models included fixed quadratic and random linear effects of age for risky behavior, but a random linear effect of age for monitoring (although the fixed linear age slope was nonsignificant). In addition, mother's attitudes significantly predicted the intercept and linear age slope for risky behavior. Although they did not significantly predict monitoring, I have added them here to illustrate how to compute indirect effects.

Chapter 9 began with person-mean-centering and baseline-centering of monitoring as a time-varying predictor of risky behavior. Both were shown to be inadequate because they do not properly distinguish the intercept, linear age slope, and residual variance contained in the monitoring predictor, each of which could potentially relate to those of risky behavior. So the purpose of this example is to demonstrate alternative software methods of estimating models of multivariate change so that you can decide what approach (software and syntax combination) will be most optimal for your own data. See chapter 9 for the results from a directed path model very similar to 2c.

In Mplus, Model 1 as an undirected multivariate MLM via long M-SEM:

```
TITLE:   Model 1: Undirected Multivariate Change Model as Long M-SEM
DATA:    FILE = Example5b.csv;   ! Syntax in same folder as data
VARIABLE:
! List of variables in data file
  NAMES = PersonID Att12 occasion age risky mon roundage
          time att4 timesq mon3;
! Variables to be analyzed in this model
  USEVARIABLE = time timesq att4 risky mon3;
  MISSING ARE ALL (-999);   ! Missing data identifier
! MLM options
  CLUSTER = PersonID;       ! Level-2 ID
  BETWEEN = att4;           ! Observed ONLY level-2 predictors
  WITHIN  = time timesq;    ! Observed ONLY level-1 predictors

ANALYSIS:    TYPE = TWOLEVEL RANDOM; ESTIMATOR = ML;

MODEL:   ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar);  ! L1 R: WP residual variances (labels)
 Rlin  | risky ON time;         ! Placeholder for R linear change
 Rquad | risky ON timesq;       ! Placeholder for R quadratic change
 Mlin  | mon3  ON time;         ! Placeholder for M linear change
 risky WITH mon3 (ResCov);      ! L1 R: WP residual covariance

%BETWEEN%
[risky mon3 Rlin Rquad Mlin];   ! Fixed intercepts, fixed change slopes
 risky mon3 (Rintvar Mintvar);  ! L2 G: Random intercept variances (labels)
 Rlin Mlin  (Rlinvar Mlinvar);  ! L2 G: Random linear change variances
 Rquad@0;                       ! No quadratic change variance

 risky Rlin ON att4;            ! Att-> R int and R linear change
  mon3 Mlin ON att4;            ! Att-> M int and M linear change
 risky WITH Rlin  (RIntLin);    ! L2 G: R Int-change covariance (label)
 mon3  WITH Mlin  (MIntLin);    ! L2 G: M Int-change covariance (label)

 risky WITH mon3  (IntCov);     ! L2 G: Random int-int covariance
 Rlin  WITH Mlin  (LinCov);     ! L2 G: Random change-change covariance
 mon3  WITH Rlin  (Int2Lin);    ! L2 G: M int, R change covariance
 Mlin  WITH risky (Lin2Int);    ! L2 G: M change, R int covariance

MODEL CONSTRAINT:  ! Linear combinations of any parameter
! First need to name each new combination
NEW(ResCor IntCor LinCor RIScor MIScor I2SCor S2ICor);

! Estimating correlations found in RCORR and GCORR
! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov  / (SQRT(Rresvar)*SQRT(Mresvar));
  IntCor = IntCov  / (SQRT(Rintvar)*SQRT(Mintvar));
  LinCor = LinCov  / (SQRT(Rlinvar)*SQRT(Mlinvar));
  RIScor = RIntLin / (SQRT(Rintvar)*SQRT(Rlinvar));
  MIScor = MIntLin / (SQRT(Mintvar)*SQRT(Mlinvar));
  I2Scor = Int2Lin / (SQRT(Mintvar)*SQRT(Rlinvar));
  S2Icor = Lin2Int / (SQRT(Mlinvar)*SQRT(Rintvar));
```

```
Number of Free Parameters                       22
Loglikelihood
          H0 Value                       -4391.884
Information Criteria
          Akaike (AIC)                    8827.768
          Bayesian (BIC)                  8943.141
          Sample-Size Adjusted BIC        8873.255
            (n* = (n + 2) / 24)
```

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
RISKY WITH				
MON3	0.287	0.028	10.441	0.000
Residual Variances				
RISKY	8.352	0.374	22.351	0.000
MON3	0.081	0.004	22.354	0.000
Between Level				
RLIN ON				
ATT4	-0.518	0.104	-4.963	0.000
MLIN ON				
ATT4	0.003	0.014	0.240	0.810
RISKY ON				
ATT4	-3.160	0.551	-5.737	0.000
MON3 ON				
ATT4	-0.044	0.057	-0.779	0.436
RISKY WITH				
RLIN	1.878	0.356	5.273	0.000
MLIN	0.040	0.039	1.044	0.296
MON3 WITH				
MLIN	0.000	0.004	-0.105	0.916
RLIN	-0.106	0.031	-3.449	0.001
RLIN WITH				
MLIN	-0.018	0.007	-2.478	0.013
RISKY WITH				
MON3	-0.853	0.168	-5.076	0.000
Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
RISKY	23.322	0.348	67.075	0.000
MON3	0.063	0.034	1.839	0.066
RLIN	1.975	0.138	14.259	0.000
MLIN	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY	18.049	2.202	8.198	0.000
MON3	0.195	0.023	8.371	0.000
RLIN	0.484	0.080	6.071	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	**0.350**	**0.028**	**12.607**	**0.000**
INTCOR	**-0.455**	**0.074**	**-6.119**	**0.000**
LINCOR	**-0.255**	**0.103**	**-2.483**	**0.013**
RISCOR	0.635	0.057	11.088	0.000
MISCOR	-0.009	0.089	-0.105	0.917
I2SCOR	-0.346	0.095	-3.646	0.000
S2ICOR	0.093	0.087	1.066	0.286

In Mplus, the same Model 1 as an undirected single-level wide SEM:

```
TITLE:   Model 1: Undirected Multivariate Change Model as Wide Single-Level SEM
DATA:    FILE = Example5b.csv;    ! Syntax in same folder as data
! Unstacking to wide format
DATA LONGTOWIDE:
! Names of old stacked former variables (without numbers)
  LONG = risky|mon3|time;
! Names of new wide variables (that use numbers)
  WIDE = risky12-risky18|mon12-mon18|age12-age18;
! Variable with level-2 ID info
  IDVARIABLE = PersonID;
! Old level-1 identifier
  REPETITION = roundage (12 13 14 15 16 17 18);
VARIABLE:
! List of variables in original data file
  NAMES = PersonID Att12 occasion age risky mon roundage
          time att4 timesq mon3;
! Variables to be analyzed in this model
  USEVARIABLE = att4 age12-age18 mon12-mon18 risky12-risky18;
  MISSING ARE ALL (-999);    ! Missing data identifier
  TSCORES = age12-age18;     ! Exact time indicator
ANALYSIS:    TYPE = RANDOM; ESTIMATOR = ML; MODEL = NOCOVARIANCES;
MODEL:   ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0
 risky12-risky18 (Rresvar);        ! L1 R: R WP residual variances held equal
 mon12-mon18     (Mresvar);        ! L1 R: M WP residual variances held equal

! Risky behavior quadratic change model using exact age as loadings
  Rint Rlin Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear change model using exact age as loadings
  Mint Mlin | mon12-mon18 AT age12-age18;
! Fixed intercept and change effects for R and M
  [Rint Rlin Rquad Mint Mlin];
! L2 G: Random int and linear change variances, no quad change variance
  Rint Rlin Mint Mlin (Rintvar Rlinvar Mintvar Mlinvar); Rquad@0;
! L2 G: Within-variable random int-change covariances for R, M
  Rint WITH Rlin (Rintlin); Mint WITH Mlin (Mintlin);
! Attitudes --> R int, R linear change, M int, M linear change
  Rint Rlin Mint Mlin ON att4;
! L2 G: Covariances between outcomes
  Rint WITH Mint (IntCov);       ! L2 G: Random int-int covariance
  Rlin WITH Mlin (LinCov);       ! L2 G: Random change-change covariance
  Mint WITH Rlin (Int2Lin);      ! L2 G: M int, R change covariance
  Mlin WITH Rint (Lin2Int);      ! L2 G: M change, R int covariance
! L1 R: WP residual covariance between same ages, held equal across age
  risky12-risky18 PWITH mon12-mon18 (ResCov);

MODEL CONSTRAINT:            ! Linear combinations of any parameter
NEW(ResCor IntCor LinCor);   ! First need to name each new created effect
! Estimating correlations found in RCORR and GCORR
! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov  / (SQRT(Rresvar)*SQRT(Mresvar));
  IntCor = IntCov  / (SQRT(Rintvar)*SQRT(Mintvar));
  LinCor = LinCov  / (SQRT(Rlinvar)*SQRT(Mlinvar));
```

MODEL = NOCOVARIANCES removes all default covariances. This will be especially useful for wide-format SEMs with "**structured residuals**" (later in this example)

Quadratic fixed effect for risky outcome not shown in diagram… and "slope" factors refer to age change (since there are many slopes!)



For balanced time, a linear change model would look like this instead (add Mquad as third variable before |):

```
Mint Mlin | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL FIT INFORMATION

Number of Free Parameters	22
Loglikelihood	
H0 Value	-4391.884
Information Criteria	
Akaike (AIC)	8827.768
Bayesian (BIC)	8900.331
Sample-Size Adjusted BIC (n* = (n + 2) / 24)	8830.633

MODEL RESULTS

		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
RINT	ON				
ATT4		-3.160	0.551	-5.737	0.000
RLIN	ON				
ATT4		-0.518	0.104	-4.963	0.000
MINT	ON				
ATT4		-0.044	0.057	-0.779	0.436
MLIN	ON				
ATT4		0.003	0.014	0.240	0.810
RINT	WITH				
RLIN		1.878	0.356	5.273	0.000
MINT		-0.853	0.168	-5.076	0.000
MLIN		0.040	0.039	1.044	0.296
MINT	WITH				
MLIN		0.000	0.004	-0.105	0.916
RLIN		-0.106	0.031	-3.449	0.001
RLIN	WITH				
MLIN		-0.018	0.007	-2.478	0.013
RISKY12	WITH				
MON12		0.287	0.028	10.441	0.000
RISKY13	WITH				
MON13		0.287	0.028	10.441	0.000
RISKY14	WITH				
MON14		0.287	0.028	10.441	0.000
RISKY15	WITH				
MON15		0.287	0.028	10.441	0.000
RISKY16	WITH				
MON16		0.287	0.028	10.441	0.000
RISKY17	WITH				
MON17		0.287	0.028	10.441	0.000
RISKY18	WITH				
MON18		0.287	0.028	10.441	0.000

Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
MON12	0.000	0.000	999.000	999.000
MON13	0.000	0.000	999.000	999.000
MON14	0.000	0.000	999.000	999.000
MON15	0.000	0.000	999.000	999.000
MON16	0.000	0.000	999.000	999.000
MON17	0.000	0.000	999.000	999.000
MON18	0.000	0.000	999.000	999.000
RISKY12	0.000	0.000	999.000	999.000
RISKY13	0.000	0.000	999.000	999.000
RISKY14	0.000	0.000	999.000	999.000
RISKY15	0.000	0.000	999.000	999.000
RISKY16	0.000	0.000	999.000	999.000
RISKY17	0.000	0.000	999.000	999.000
RISKY18	0.000	0.000	999.000	999.000
RINT	23.322	0.348	67.074	0.000
RLIN	1.975	0.138	14.259	0.000
MINT	0.063	0.034	1.839	0.066
MLIN	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
MON12	0.081	0.004	22.354	0.000
MON13	0.081	0.004	22.354	0.000
MON14	0.081	0.004	22.354	0.000
MON15	0.081	0.004	22.354	0.000
MON16	0.081	0.004	22.354	0.000
MON17	0.081	0.004	22.354	0.000
MON18	0.081	0.004	22.354	0.000
RISKY12	8.352	0.374	22.351	0.000
RISKY13	8.352	0.374	22.351	0.000
RISKY14	8.352	0.374	22.351	0.000
RISKY15	8.352	0.374	22.351	0.000
RISKY16	8.352	0.374	22.351	0.000
RISKY17	8.352	0.374	22.351	0.000
RISKY18	8.352	0.374	22.351	0.000
RINT	18.049	2.202	8.198	0.000
RLIN	0.484	0.080	6.071	0.000
MINT	0.195	0.023	8.371	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	**0.350**	**0.028**	**12.607**	**0.000**
INTCOR	**-0.455**	**0.074**	**-6.119**	**0.000**
LINCOR	**-0.255**	**0.103**	**-2.483**	**0.013**

Model 2a: Partially Directed Path Multivariate MLM in Mplus as Long M-SEM: Monitor → Risky for BP intercepts and slopes, but WP residuals covary

```
TITLE:  Model 2a: Partially Directed Multivariate Change Model as Long M-SEM
        L1 WP relation still as residual covariance

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 MLM )

MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar); ! L1 R: WP residual variances (labels)
 Rlin  | risky ON time;        ! Placeholder for R linear change
 Rquad | risky ON timesq;      ! Placeholder for R quadratic change
 Mlin  | mon3  ON time;        ! Placeholder for M linear change
 risky WITH mon3 (ResCov);     ! L1 R: Still WP residual covariance

%BETWEEN%
[risky mon3 Rlin Rquad Mlin];  ! Fixed intercepts, fixed change slopes
 risky mon3 (Rintvar Mintvar); ! L2 G: Random intercept variances (labels)
 Rlin Mlin  (Rlinvar Mlinvar); ! L2 G: Random linear change variances
 Rquad@0;                      ! No quadratic change variance
 risky Rlin ON att4;           ! Att-> R int, R linear change
  mon3 Mlin ON att4;           ! Att-> M int, M linear change
 risky WITH Rlin (RIntLin);    ! L2 G: R int-change covariance (label)
 mon3  WITH Mlin (MIntLin);    ! L2 G: M int-change covariance (label)

! Although we have changed the int-int and change-change relations to direct
! paths from M -> R instead of covariances, they still represent total L2
! BP relationships because the L1 relationship is still a covariance

 risky ON   mon3 (BPIntEff);    ! L2 BP int-to-int effect
 Rlin  ON   Mlin (BPLinEff);    ! L2 BP change-to-change effect

 mon3  WITH Rlin  (Int2Lin);    ! L2 G: M int, R change covariance
 Mlin  WITH risky (Lin2Int);    ! L2 G: M change, R int covariance

MODEL CONSTRAINT:
! All values below are variances from undirected model 1
NEW(ResCor IntStd LinStd);

! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov / (SQRT(8.3538)*SQRT(0.08077));    ! L1 WP res corr

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP int-int effect
  LinStd = BPLinEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP change-change
```

This is an equivalent model, just with a different way of specifying the level-2 BP intercept-to-intercept and change-to-change relationships.

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
RISKY WITH				
MON3	0.287	0.028	10.441	0.000
Residual Variances				
RISKY	8.352	0.374	22.351	0.000
MON3	0.081	0.004	22.354	0.000
Between Level **– changed parameters are in BOLD**				
RLIN ON				
MLIN	**-1.736**	**0.713**	**-2.434**	**0.015**
RLIN ON				
ATT4	-0.512	0.105	-4.869	0.000
MLIN ON				
ATT4	0.003	0.014	0.240	0.810
RISKY ON				
ATT4	-3.354	0.528	-6.348	0.000
MON3	**-4.380**	**0.797**	**-5.497**	**0.000**
MON3 ON				
ATT4	-0.044	0.057	-0.779	0.436
RISKY WITH				
RLIN	1.480	0.345	4.285	0.000
MLIN	0.039	0.038	1.023	0.306
MON3 WITH				
MLIN	0.000	0.004	-0.105	0.916
RLIN	-0.107	0.031	-3.454	0.001
Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
RISKY	23.598	0.338	69.837	0.000
MON3	0.063	0.034	1.839	0.066
RLIN	1.969	0.139	14.195	0.000
MLIN	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY	14.315	2.030	7.053	0.000
MON3	0.195	0.023	8.371	0.000
RLIN	0.453	0.081	5.564	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	0.350	0.034	10.441	0.000
INTSTD	-0.455	0.083	-5.497	0.000
LINSTD	-0.254	0.104	-2.434	0.015

In Mplus, the same Model 2a as a partially directed wide single-level SEM: Monitor → Risky for intercepts and slopes, but residuals covary

```
TITLE:   Model 2a: Partially Directed Multivariate Change Model as Wide
         Single-Level SEM, L1 WP relation still as residual covariance

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 SEM )

MODEL:   ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0
 risky12-risky18 (Rresvar);        ! L1 R: WP R residual variances held equal
 mon12-mon18     (Mresvar);        ! L1 R: WP M residual variances held equal

! Risky behavior quadratic change model using exact age as loadings
  Rint Rlin Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear change model using exact age as loadings
  Mint Mlin | mon12-mon18 AT age12-age18;
! Fixed intercept and change effects for R and M
  [Rint Rlin Rquad Mint Mlin];
! L2 G: Random int and linear change variances, no quad change variance
  Rint Rlin Mint Mlin (Rintvar Rlinvar Mintvar Mlinvar); Rquad@0;
! L2 G: Within-variable random int-change covariances for R, M
  Rint WITH Rlin (Rintlin); Mint WITH Mlin (Mintlin);
! Attitudes --> R int, R linear change, M int, M change slope
  Rint Rlin Mint Mlin ON att4;

! Although we have changed the int-int and change-change relations to direct
! paths from M -> R instead of covariances, they still represent total L2
! BP relationships because the L1 relationship is still a covariance

  Rint ON Mint   (BPIntEff);     ! L2 BP int-to-int effect
  Rlin ON Mlin   (BPLinEff);     ! L2 BP change-to-change effect

  Mint WITH Rlin (Int2Lin);      ! L2 G: M int, R change covariance
  Mlin WITH Rint (Lin2Int);      ! L2 G: M change, R int covariance

! L1 R: WP residual covariance between same ages, held equal across age
  risky12-risky18 PWITH mon12-mon18 (ResCov);

MODEL CONSTRAINT:  ! All values below are variances from undirected model 1
  NEW(ResCor IntStd LinStd);

! Corr = Cov / (SQRT(Yvar)*SQRT(Xvar))
  ResCor = ResCov / (SQRT(8.3538)*SQRT(0.08077));    ! L1 WP res corr

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP int-int effect
  LinStd = BPLinEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP change-change
```

This is an equivalent model, just with a different way of specifying the level-2 BP intercept-to-intercept and change-to-change relationships.

Quadratic fixed effect for risky outcome not shown in diagram… and "slope" factors refer to age change (since there are many slopes!)



For balanced time, a linear change model would look like this instead (add Mquad as third variable before |):

```
Mint Mlin | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL RESULTS **– changed parameters are in BOLD**

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
RINT ON				
MINT	**-4.380**	**0.797**	**-5.496**	**0.000**
RLIN ON				
MLIN	**-1.735**	**0.713**	**-2.434**	**0.015**
RINT ON				
ATT4	-3.354	0.528	-6.348	0.000
RLIN ON				
ATT4	-0.512	0.105	-4.869	0.000
MINT ON				
ATT4	-0.044	0.057	-0.779	0.436
MLIN ON				
ATT4	0.003	0.014	0.240	0.810
RINT WITH				
RLIN	1.480	0.345	4.285	0.000
MLIN	0.039	0.038	1.023	0.306
MINT WITH				
MLIN	0.000	0.004	-0.105	0.916
RLIN	-0.107	0.031	-3.454	0.001
RISKY12 WITH				
MON12	0.287	0.028	10.441	0.000
RISKY13 WITH				
MON13	0.287	0.028	10.441	0.000
RISKY14 WITH				
MON14	0.287	0.028	10.441	0.000
RISKY15 WITH				
MON15	0.287	0.028	10.441	0.000
RISKY16 WITH				
MON16	0.287	0.028	10.441	0.000
RISKY17 WITH				
MON17	0.287	0.028	10.441	0.000
RISKY18 WITH				
MON18	0.287	0.028	10.441	0.000

Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
RISKY12	0.000	0.000	999.000	999.000
RISKY13	0.000	0.000	999.000	999.000
RISKY14	0.000	0.000	999.000	999.000
RISKY15	0.000	0.000	999.000	999.000
RISKY16	0.000	0.000	999.000	999.000
RISKY17	0.000	0.000	999.000	999.000
RISKY18	0.000	0.000	999.000	999.000
MON12	0.000	0.000	999.000	999.000
MON13	0.000	0.000	999.000	999.000
MON14	0.000	0.000	999.000	999.000
MON15	0.000	0.000	999.000	999.000
MON16	0.000	0.000	999.000	999.000
MON17	0.000	0.000	999.000	999.000
MON18	0.000	0.000	999.000	999.000
RINT	23.598	0.338	69.836	0.000
RLIN	1.969	0.139	14.195	0.000
MINT	0.063	0.034	1.839	0.066
MLIN	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY12	8.352	0.374	22.351	0.000
RISKY13	8.352	0.374	22.351	0.000
RISKY14	8.352	0.374	22.351	0.000
RISKY15	8.352	0.374	22.351	0.000
RISKY16	8.352	0.374	22.351	0.000
RISKY17	8.352	0.374	22.351	0.000
RISKY18	8.352	0.374	22.351	0.000
MON12	0.081	0.004	22.354	0.000
MON13	0.081	0.004	22.354	0.000
MON14	0.081	0.004	22.354	0.000
MON15	0.081	0.004	22.354	0.000
MON16	0.081	0.004	22.354	0.000
MON17	0.081	0.004	22.354	0.000
MON18	0.081	0.004	22.354	0.000
RINT	14.315	2.030	7.053	0.000
RLIN	0.453	0.081	5.564	0.000
MINT	0.195	0.023	8.371	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESCOR	0.350	0.034	10.441	0.000
INTSTD	-0.455	0.083	-5.496	0.000
LINSTD	-0.254	0.104	-2.434	0.015

Model 2b: Partially Directed Path Multivariate MLM in Mplus as Long M-SEM: Monitor → Risky for WP residuals within L1 model

Also demonstrating how to request BP indirect effects

```
TITLE:   Model 2b: Partially Directed Multivariate Change Model as Long M-SEM
         L1 WP relation NOW as direct path specified in L1 WITHIN

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 MLM )

MODEL:   ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar); ! L1 R: Residual variances (labels)
 Rlin  | risky ON time;        ! Placeholder for R linear change
 Rquad | risky ON timesq;      ! Placeholder for R quadratic change
 Mlin  | mon3  ON time;        ! Placeholder for M linear change
 risky ON mon3 (ResEff);       ! L1 WP fixed effect M->R here (label)

%BETWEEN%
[risky mon3 Rlin Rquad Mlin];  ! Fixed intercepts, fixed change slopes
 risky mon3 (Rintvar Mintvar); ! L2 G: Random intercept variances (labels)
 Rlin Mlin  (Rlinvar Mlinvar); ! L2 G: Random linear change variances
 Rquad@0;                      ! No quadratic change variance
 risky Rlin ON att4 (XtoYint XtoYlin);  ! Att-> R int, R linear change
  mon3 Mlin ON att4 (XtoMint XtoMlin);  ! Att-> M int, M linear change
 risky WITH Rlin (RIntLin);     ! L2 G: R int-change covariance (label)
  mon3 WITH Mlin (MIntLin);     ! L2 G: M int-change covariance (label)

! Although the intercept -> intercept path remains the total L2 BP effect,
! now the change -> change path becomes the L2 contextual effect instead

 risky ON   mon3  (BPIntEff);   ! STILL L2 BP int-to-int effect
 Rlin  ON   Mlin  (LinCont);    ! NOW L2 contextual change-to-change effect

 mon3  WITH Rlin  (Int2Lin);    ! L2 G: M int, R change covariance
 Mlin  WITH risky (Lin2Int);    ! L2 G: M change, R int covariance

MODEL CONSTRAINT:
NEW(ResStd IntStd BPLinEff LinStd indBPint indBPlin);

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStd = ResEff   * SQRT(0.08077) / SQRT(8.3538); ! STD L1 WP res-res effect
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP int-int effect
  BPLinEff = ResEff + LinCont;            ! WP + Context = BP change-change
  LinStd = BPLinEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP change effect
  indBPint = XtoMint * BPIntEff;             ! BP intercept indirect effect
  indBPlin = XtoMlin * BPLinEff;             ! BP change indirect effect
```

This is still an equivalent model, just with a different way of specifying the level-1 WP residual-to-residual relationship. This syntax method will only work for level-1 effects that are only fixed, though—adding random effects and/or cross-level interactions requires the level-1 placeholder syntax.

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level – **changed parameters are in BOLD**				
RISKY ON				
MON3	**3.559**	**0.301**	**11.809**	**0.000**
Residual Variances				
RISKY	7.329	0.328	22.353	0.000
MON3	0.081	0.004	22.354	0.000
Between Level – **changed parameters are in BOLD**				
RLIN ON				
MLIN	**-5.294**	**0.806**	**-6.569**	**0.000**
RLIN ON				
ATT4	-0.512	0.105	-4.869	0.000
MLIN ON				
ATT4	0.003	0.014	0.240	0.810
RISKY ON				
ATT4	-3.354	0.528	-6.348	0.000
MON3	-4.380	0.797	-5.497	0.000
MON3 ON				
ATT4	-0.044	0.057	-0.779	0.436
RISKY WITH				
RLIN	1.480	0.345	4.285	0.000
MLIN	0.039	0.038	1.023	0.306
MON3 WITH				
MLIN	0.000	0.004	-0.105	0.916
RLIN	-0.107	0.031	-3.454	0.001
Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
RISKY	23.598	0.338	69.837	0.000
MON3	0.063	0.034	1.839	0.066
RLIN	1.969	0.139	14.195	0.000
MLIN	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY	14.315	2.030	7.053	0.000
MON3	0.195	0.023	8.371	0.000
RLIN	0.453	0.081	5.564	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESSTD	0.350	0.034	10.441	0.000
INTSTD	-0.455	0.083	-5.497	0.000
BPLINEFF	**-1.736**	**0.713**	**-2.434**	**0.015**
LINSTD	-0.254	0.104	-2.434	0.015
INDBPINT	0.194	0.251	0.772	0.440
INDBPLIN	-0.006	0.024	-0.239	0.811

In Mplus, the same Model 2b as a partially directed wide single-level SEM: Monitor → Risky for WP residuals using structured residuals

```
TITLE:   Model 2b: Partially Directed Multivariate Change Model as Wide
         Single-Level SEM, L1 WP effect as direct path via structured residuals
( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 SEM )
MODEL:   ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0

! Risky behavior quadratic change model using exact age as loadings
  Rint Rlin Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear change model using exact age as loadings
  Mint Mlin | mon12-mon18 AT age12-age18;
! Fixed intercept and change effects for R and M
  [Rint Rlin Rquad Mint Mlin];
! L2 G: Random int and linear change variances, no quad change variance
  Rint Rlin Mint Mlin (Rintvar Rlinvar Mintvar Mlinvar); Rquad@0;
! L2 G: Within-variable random int-change covariances for R, M
  Rint WITH Rlin (Rintlin); Mint WITH Mlin (Mintlin);
! Attitudes --> R int, R change, M int, M change
  Rint Rlin Mint Mlin ON att4 (XtoYint XtoYlin XtoMint XtoMlin);
  Rint ON Mint   (BPIntEff);     ! BP int-to-int effect
  Rlin ON Mlin   (BPLinEff);     ! BP change-to-change effect
  Mint WITH Rlin (Int2Lin);      ! L2 G: M int, R change covariance
  Mlin WITH Rint (Lin2Int);      ! L2 G: M change, R int covariance

! Define new latent factors for residuals at each occasion
  Frisky12 BY risky12@1; Frisky13 BY risky13@1; Frisky14 BY risky14@1;
  Frisky15 BY risky15@1; Frisky16 BY risky16@1; Frisky17 BY risky17@1;
  Frisky18 BY risky18@1; Fmon12 BY mon12@1; Fmon13 BY mon13@1;
  Fmon14 BY mon14@1; Fmon15 BY mon15@1; Fmon16 BY mon16@1;
  Fmon17 BY mon17@1; Fmon18 BY mon18@1;
! All factor means fixed to 0
  [Frisky12-Frisky18@0 Fmon12-Fmon18@0];
! Shut off old residual variances
  risky12-risky18@0 mon12-mon18@0;
! Hold new residual variances equal over time
  Frisky12-Frisky18 (Rresvar); ! L1 R: R WP residual variances held equal
  Fmon12-Fmon18     (Mresvar); ! L1 R: M WP residual variances held equal
! Factor residual WP effect between same ages, held equal across age
  Frisky12-Frisky18 PON Fmon12-Fmon18 (ResEff);

MODEL CONSTRAINT:
NEW(ResStd IntStd LinStd indBPint indBPlin);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStd = ResEff    * SQRT(0.08077) / SQRT(8.3538);  ! STD WP res-res effect
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP int-int effect
  LinStd = BPLinEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP change-change
  indBPint = XtoMint * BPIntEff;          ! BP intercept indirect effect
  indBPlin = XtoMlin * BPLinEff;          ! BP change indirect effect
```

For simpler residuals code using Mplus v. 8.8+, see Hoffman & Hall (2024) Example 1 (supplemental materials .zip)

This is still an equivalent model, just with a different way of specifying the level-1 WP residual-to-residual relationship. This syntax method will only work for level-1 effects that are only fixed, though…

Quadratic fixed effect for risky outcome not shown in diagram… and "slope" factors refer to age change (since there are many slopes!)



For balanced time, a linear change model would look like this instead (add Mquad as third variable before |):

```
Mint Mlin | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL RESULTS **– changed parameters are in BOLD**

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Factor loadings set to 1 omitted				
RINT ON				
MINT	-4.380	0.797	-5.496	0.000
RLIN ON				
MLIN	-1.736	0.713	-2.434	0.015
_FRISKY12 ON				
FMON12	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY13 ON				
FMON13	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY14 ON				
FMON14	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY15 ON				
FMON15	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY16 ON				
FMON16	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY17 ON				
FMON17	**3.563**	**0.302**	**11.810**	**0.000**
FRISKY18 ON				
FMON18	**3.563**	**0.302**	**11.810**	**0.000**
RINT ON				
ATT4	-3.354	0.528	-6.348	0.000
RLIN ON				
ATT4	-0.512	0.105	-4.869	0.000
MINT ON				
ATT4	-0.044	0.057	-0.779	0.436
MLIN ON				
ATT4	0.003	0.014	0.240	0.810
RINT WITH				
RLIN	1.480	0.345	4.285	0.000
MLIN	0.039	0.038	1.023	0.306
MINT WITH				
MLIN	0.000	0.004	-0.105	0.916
RLIN	-0.107	0.031	-3.454	0.001

Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
(Intercepts fixed to 0 omitted)				
RINT	23.598	0.338	69.836	0.000
RLIN	1.969	0.139	14.195	0.000
MINT	0.063	0.034	1.839	0.066
MLIN	-0.003	0.008	-0.380	0.704
Variances				
FMON12	0.081	0.004	22.327	0.000
FMON13	0.081	0.004	22.327	0.000
FMON14	0.081	0.004	22.327	0.000
FMON15	0.081	0.004	22.327	0.000
FMON16	0.081	0.004	22.327	0.000
FMON17	0.081	0.004	22.327	0.000
FMON18	0.081	0.004	22.327	0.000
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
(Residual variances fixed to 0 omitted)				
FRISKY12	7.328	0.328	22.349	0.000
FRISKY13	7.328	0.328	22.349	0.000
FRISKY14	7.328	0.328	22.349	0.000
FRISKY15	7.328	0.328	22.349	0.000
FRISKY16	7.328	0.328	22.349	0.000
FRISKY17	7.328	0.328	22.349	0.000
FRISKY18	7.328	0.328	22.349	0.000
RINT	14.315	2.030	7.053	0.000
RLIN	0.453	0.081	5.564	0.000
MINT	0.195	0.023	8.371	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESSTD	0.350	0.034	10.441	0.000
INTSTD	-0.455	0.083	-5.496	0.000
LINSTD	-0.254	0.104	-2.434	0.015
INDBPINT	0.194	0.251	0.772	0.440
INDBPLIN	-0.006	0.024	-0.239	0.811

Model 2c: Partially Directed Path Multivariate MLM in Mplus as Long M-SEM: Monitor → Risky for WP residuals within L2 model via placeholder syntax; also demonstrating how to request BP indirect effects

```
TITLE:  Model 2c: Partially Directed Multivariate Change Model as Long M-SEM
        L1 WP relation using placeholder specified in L2 BETWEEN
( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 MLM )

MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar); ! L1 R: WP residual variances (labels)
 Rlin  | risky ON time;        ! Placeholder for R linear change
 Rquad | risky ON timesq;      ! Placeholder for R quadratic change
 Mlin  | mon3  ON time;        ! Placeholder for M linear change
 WPres | risky ON mon3;        ! NEW placeholder for L1 WP effect M->R

%BETWEEN%
[risky mon3 Rlin Rquad Mlin];  ! Fixed intercepts, fixed change slopes
 risky mon3 (Rintvar Mintvar); ! L2 G: Random intercept variances (labels)
 Rlin Mlin  (Rlinvar Mlinvar); ! L2 G: Random linear change variances
 Rquad@0;                      ! No quadratic change variance
risky Rlin ON att4 (XtoYint XtoYlin);  ! Att-> R int, R linear change
 mon3 Mlin ON att4 (XtoMint XtoMlin);  ! Att-> M int, M linear change
 risky WITH Rlin (RIntLin);     ! L2 G: R Int-change covariance (label)
  mon3 WITH Mlin (MIntLin);     ! L2 G: M Int-change covariance (label)

! And now both the intercept -> intercept path and the change -> change path
! are L2 contextual effects given the L1 placeholder for WP M -> R effect

 risky ON   mon3  (IntCont);    ! NOW L2 contextual int-to-int effect
 Rlin  ON   Mlin  (LinCont);    ! STILL L2 contextual change-to-change effect
 mon3  WITH Rlin  (Int2Lin);    ! L2 G: M int, R change covariance
 Mlin  WITH risky (Lin2Int);    ! L2 G: M change, R int covariance

 [WPres] (ResEff);  ! Fixed effect for L1 WP M->R (as defined earlier)
  WPres@0;          ! No random L1 WP M->R slope variance

MODEL CONSTRAINT:
NEW(ResStd BPIntEff IntStd BPLinEff LinStd indBPint indBPlin);

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStd = ResEff   * SQRT(0.08077) / SQRT(8.3538);   ! STD WP effect
  BPIntEff = ResEff + IntCont;                  ! WP + Context = BP int-int
  IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644);  ! STD BP int-int effect
  BPLinEff = ResEff + LinCont;              ! WP + Context = BP change-change
  LinStd = BPLinEff * SQRT(0.01049) / SQRT(0.48830);  ! STD BP change-change
  indBPint = XtoMint * BPIntEff;             ! BP intercept indirect effect
  indBPlin = XtoMlin * BPLinEff;             ! BP change indirect effect
```

This is still an equivalent model, just with a different syntax for specifying the same level-1 WP residual-to-residual directed relationship. This is the version that is necessary in order to have the level-1 effect become random or systematically varying (i.e., add cross-level interactions). **But if you switch to Bayes estimation, then all the L2 effects are BP instead (see Hoffman 2019)!!!**

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Within Level				
Residual Variances				
RISKY	7.329	0.328	22.353	0.000
MON3	0.081	0.004	22.354	0.000
Between Level **– changed parameters are in BOLD**				
RLIN ON				
MLIN	-5.294	0.806	-6.569	0.000
RLIN ON				
ATT4	-0.512	0.105	-4.869	0.000
MLIN ON				
ATT4	0.003	0.014	0.240	0.810
RISKY ON				
ATT4	-3.354	0.528	-6.348	0.000
MON3	**-7.938**	**0.872**	**-9.099**	**0.000**
MON3 ON				
ATT4	-0.044	0.057	-0.779	0.436
RISKY WITH				
RLIN	1.480	0.345	4.285	0.000
MLIN	0.039	0.038	1.023	0.306
MON3 WITH				
MLIN	0.000	0.004	-0.105	0.916
RLIN	-0.107	0.031	-3.454	0.001
Means				
RQUAD	0.147	0.021	7.117	0.000
WPRES (here now)	**3.559**	**0.301**	**11.810**	**0.000**
Intercepts				
RISKY	23.598	0.338	69.837	0.000
MON3	0.063	0.034	1.839	0.066
RLIN	1.969	0.139	14.195	0.000
MLIN	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
WPRES	0.000	0.000	999.000	999.000
Residual Variances				
RISKY	14.315	2.030	7.053	0.000
MON3	0.195	0.023	8.371	0.000
RLIN	0.453	0.081	5.564	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESSTD	0.350	0.030	11.810	0.000
BPINTEFF	**-4.380**	**0.797**	**-5.496**	**0.000**
INTSTD	-0.455	0.083	-5.496	0.000
BPLINEFF	**-1.735**	**0.713**	**-2.434**	**0.015**
LINSTD	-0.254	0.104	-2.434	0.015
INDBPINT	**0.194**	**0.251**	**0.772**	**0.440**
INDBPLIN	**-0.006**	**0.024**	**-0.239**	**0.811**

In Mplus, the same Model 2c as a partially directed wide single-level SEM: Monitor → Risky for WP residuals using original residuals

```
TITLE:   Model 2c: Partially Directed Multivariate Change Model as Wide
         Single-Level SEM, L1 WP effect as direct path via original residuals

( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 SEM )
MODEL:   ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0
 risky12-risky18 (Rresvar);        ! L1 R: R WP residual variances held equal
 mon12-mon18     (Mresvar);        ! L1 R: M WP residual variances held equal

! Risky behavior quadratic change model using exact age as loadings
  Rint Rlin Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear change model using exact age as loadings
  Mint Mlin | mon12-mon18 AT age12-age18;
! Fixed intercept and change effects for R and M
  [Rint Rlin Rquad Mint Mlin];
! L2 G: Random int and linear change variances, no quad change variance
  Rint Rlin Mint Mlin (Rintvar Rlinvar Mintvar Mlinvar); Rquad@0;
! L2 G: Within-variable random int-change covariances for R, M
  Rint WITH Rlin (Rintlin); Mint WITH Mlin (Mintlin);
! Attitudes --> R int, R change, M int, M change
  Rint Rlin Mint Mlin ON att4 (XtoYint XtoYlin XtoMint XtoMlin);

  Rint ON Mint   (IntCont);     ! NOW L2 contextual int-to-int effect
  Rlin ON Mlin   (LinCont);     ! NOW L2 contextual change-to-change effect
  Mint WITH Rlin (Int2Lin);     ! L2 G: M int, R change covariance
  Mlin WITH Rint (Lin2Int);     ! L2 G: M change, R int covariance

! L1 WP M -> R slope between same ages, held equal across age
  risky12-risky18 PON mon12-mon18 (ResEff);

MODEL CONSTRAINT:
NEW(ResStd BPIntEff IntStd BPLinEff LinStd indBPint indBPlin);

! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
ResStd = ResEff   * SQRT(0.08077) / SQRT(8.3538);  ! STD WP effect
BPIntEff = ResEff + IntCont;                       ! WP + Context = BP int-int
IntStd = BPIntEff * SQRT(0.19530) / SQRT(18.0644); ! STD BP int-int effect
BPLinEff = ResEff + LinCont;                       ! WP + Context = BP change-change
LinStd = BPLinEff * SQRT(0.01049) / SQRT(0.48830); ! STD BP change-change
indBPint = XtoMint * BPIntEff;                     ! BP intercept indirect effect
indBPlin = XtoMlin * BPLinEff;                     ! BP change indirect effect
```

This is still an equivalent model, just with a different syntax for specifying the same level-1 WP directed relationship. The consequence is that the intercept-to-intercept and change-to-change relationships become L2 contextual effects (as in the MLM version). Oddly, if we were to switch to ON for the intercept–change cross-variable relationships, those stay L2 BP (see chapter 9 for an example using this version of the model).

Quadratic fixed effect for risky outcome not shown in diagram… and "slope" factors refer to age change (since there are many slopes!)




```
For balanced time, a linear change model would look like
this instead (add Mquad as third variable before |):

Mint Mlin | mon12@-6 mon13@-5 mon14@-4 mon15@-3
            mon16@-2 mon17@-1 mon18@0;
```

MODEL RESULTS **– changed parameters are in BOLD**

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
Factor loadings set to 1 omitted				
RINT ON				
MINT	**-7.939**	**0.872**	**-9.099**	**0.000**
RLIN ON				
MLIN	**-5.294**	**0.806**	**-6.569**	**0.000**
RINT ON				
ATT4	-3.354	0.528	-6.348	0.000
RLIN ON				
ATT4	-0.512	0.105	-4.869	0.000
MINT ON				
ATT4	-0.044	0.057	-0.779	0.436
MLIN ON				
ATT4	0.003	0.014	0.240	0.810
RISKY12 ON				
MON12	3.559	0.301	11.810	0.000
RISKY13 ON				
MON13	3.559	0.301	11.810	0.000
RISKY14 ON				
MON14	3.559	0.301	11.810	0.000
RISKY15 ON				
MON15	3.559	0.301	11.810	0.000
RISKY16 ON				
MON16	3.559	0.301	11.810	0.000
RISKY17 ON				
MON17	3.559	0.301	11.810	0.000
RISKY18 ON				
MON18	3.559	0.301	11.810	0.000
RINT WITH				
RLIN	1.480	0.345	4.285	0.000
MLIN	0.039	0.038	1.023	0.306
MINT WITH				
MLIN	0.000	0.004	-0.105	0.916
RLIN	-0.107	0.031	-3.454	0.001
Means				
RQUAD	0.147	0.021	7.117	0.000
Intercepts				
Intercepts fixed to 0 omitted				
RINT	23.598	0.338	69.836	0.000
RLIN	1.969	0.139	14.195	0.000
MINT	0.063	0.034	1.839	0.066
MLIN	-0.003	0.008	-0.380	0.704
Variances				
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
RISKY12	7.329	0.328	22.353	0.000
RISKY13	7.329	0.328	22.353	0.000
RISKY14	7.329	0.328	22.353	0.000
RISKY15	7.329	0.328	22.353	0.000
RISKY16	7.329	0.328	22.353	0.000
RISKY17	7.329	0.328	22.353	0.000
RISKY18	7.329	0.328	22.353	0.000
MON12	0.081	0.004	22.354	0.000
MON13	0.081	0.004	22.354	0.000
MON14	0.081	0.004	22.354	0.000
MON15	0.081	0.004	22.354	0.000
MON16	0.081	0.004	22.354	0.000
MON17	0.081	0.004	22.354	0.000
MON18	0.081	0.004	22.354	0.000
RINT	14.315	2.030	7.053	0.000
RLIN	0.453	0.081	5.564	0.000
MINT	0.195	0.023	8.371	0.000
MLIN	0.010	0.001	7.802	0.000
New/Additional Parameters				
RESSTD	0.350	0.030	11.810	0.000
BPINTEFF	**-4.380**	**0.797**	**-5.496**	**0.000**
INTSTD	-0.455	0.083	-5.496	0.000
BPLINEFF	**-1.735**	**0.713**	**-2.434**	**0.015**
LINSTD	-0.254	0.104	-2.434	0.015
INDBPINT	0.194	0.251	0.772	0.440
INDBPLIN	-0.006	0.024	-0.239	0.811

Here is an example of how to use "structured residuals" to fit two cross-lagged effects at level 1 using wide single-level SEM:
Model 3a, which switches to covariances at level 2 when fitting these models (per convention, to be agnostic as to which comes first)

```
TITLE:  Model 3a: Wide SEM Structured Residuals to Fit 2 Cross-Lagged Paths
  ( DATA, VARIABLE, and ANALYSIS are the same as for Model 1 SEM )

MODEL:  ! R = risky behavior, M = monitoring
[risky12-risky18@0 mon12-mon18@0]; ! All variable intercepts fixed to 0

! Risky behavior quadratic change model using exact age as loadings
  Rint Rlin Rquad | risky12-risky18 AT age12-age18;
! Monitoring linear change model using exact age as loadings
  Mint Mlin | mon12-mon18 AT age12-age18;
! Fixed intercept and change effects for R and M
  [Rint Rlin Rquad Mint Mlin];
! L2 G: Random int and linear change variances, no quad change variance
  Rint Rlin Mint Mlin (Rintvar Rlinvar Mintvar Mlinvar); Rquad@0;
! Attitudes --> R int, R change, M int, M change
  Rint Rlin Mint Mlin ON att4 (XtoYint XtoYlin XtoMint XtoMlin);
! L2 G: covariances for random intercepts and slopes across outcomes
  Rint Rlin Mint Mlin WITH Rint Rlin Mint Mlin;

! Define new latent factors for residuals at each occasion
  Frisky12 BY risky12@1; Frisky13 BY risky13@1; Frisky14 BY risky14@1;
  Frisky15 BY risky15@1; Frisky16 BY risky16@1; Frisky17 BY risky17@1;
  Frisky18 BY risky18@1; Fmon12 BY mon12@1; Fmon13 BY mon13@1;
  Fmon14 BY mon14@1; Fmon15 BY mon15@1; Fmon16 BY mon16@1;
  Fmon17 BY mon17@1; Fmon18 BY mon18@1;
! All factor means fixed to 0
  [Frisky12-Frisky18@0 Fmon12-Fmon18@0];
! Shut off old residual variances
  risky12-risky18@0 mon12-mon18@0;
! Hold new residual variances equal over time if predicted
  Frisky13-Frisky18 (Rresvar); ! L1 R: R WP residual variances held equal
  Fmon13-Fmon18     (Mresvar); ! L1 R: M WP residual variances held equal

! WP residual covariance for unpredicted occasion between same age
  Frisky12 WITH Fmon12;
! WP residual covariance for predicted occasions held equal across age
  Frisky13-Frisky18 PWITH Fmon13-Fmon18 (ResCov);

! Cross-lagged WP effects predicting next occasion, held equal across age
  Frisky13-Frisky18 PON Fmon12-Fmon17     (MR2RR);
  Fmon13-Fmon18     PON Frisky12-Frisky17 (RR2MR);

MODEL CONSTRAINT:
NEW(MR2RRsd RR2MRsd);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  MR2RRsd = MR2RR * SQRT(0.08077) / SQRT(8.3538);  ! STD M->R lagged effect
  RR2MRsd = RR2MR * SQRT(8.3538)  / SQRT(0.08077); ! STD R->M lagged effect
```

This is NOT an equivalent model given the two new cross-lagged effects, along with the two separate residual variances and covariance for age 12.

MODEL RESULTS – Parameters fixed to 0 or 1 are omitted for brevity
changed parameters in bold

		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
FRISKY13	**ON**				
FMON12		**-0.171**	**0.368**	**-0.465**	**0.642**
FRISKY14	**ON**				
FMON13		**-0.171**	**0.368**	**-0.465**	**0.642**
FRISKY15	**ON**				
FMON14		**-0.171**	**0.368**	**-0.465**	**0.642**
FRISKY16	**ON**				
FMON15		**-0.171**	**0.368**	**-0.465**	**0.642**
FRISKY17	**ON**				
FMON16		**-0.171**	**0.368**	**-0.465**	**0.642**
FRISKY18	**ON**				
FMON17		**-0.171**	**0.368**	**-0.465**	**0.642**
FMON13	**ON**				
FRISKY12		**0.008**	**0.004**	**2.131**	**0.033**
FMON14	**ON**				
FRISKY13		**0.008**	**0.004**	**2.131**	**0.033**
FMON15	**ON**				
FRISKY14		**0.008**	**0.004**	**2.131**	**0.033**
FMON16	**ON**				
FRISKY15		**0.008**	**0.004**	**2.131**	**0.033**
FMON17	**ON**				
FRISKY16		**0.008**	**0.004**	**2.131**	**0.033**
FMON18	**ON**				
FRISKY17		**0.008**	**0.004**	**2.131**	**0.033**
RINT	ON				
ATT4		-3.153	0.552	-5.713	0.000
RLIN	ON				
ATT4		-0.514	0.105	-4.877	0.000
MINT	ON				
ATT4		-0.043	0.057	-0.765	0.444
MLIN	ON				
ATT4		0.004	0.014	0.309	0.757
RINT	WITH				
RLIN		1.945	0.363	5.365	0.000
MINT		-0.884	0.170	-5.213	0.000
MLIN		0.038	0.039	0.976	0.329
RLIN	WITH				
MINT		-0.110	0.031	-3.511	0.000
MLIN		-0.018	0.008	-2.375	0.018
MINT	WITH				
MLIN		-0.001	0.004	-0.310	0.756

FRISKY12 WITH				
FMON12	0.053	0.087	0.608	0.543
FRISKY13 WITH				
FMON13	0.329	0.033	9.970	0.000
FRISKY14 WITH				
FMON14	0.329	0.033	9.970	0.000
FRISKY15 WITH				
FMON15	0.329	0.033	9.970	0.000
FRISKY16 WITH				
FMON16	0.329	0.033	9.970	0.000
FRISKY17 WITH				
FMON17	0.329	0.033	9.970	0.000
FRISKY18 WITH				
FMON18	0.329	0.033	9.970	0.000
Means				
RQUAD	0.145	0.021	7.041	0.000
Intercepts				
RINT	23.322	0.347	67.162	0.000
RLIN	1.971	0.137	14.378	0.000
MINT	0.064	0.034	1.873	0.061
MLIN	-0.003	0.008	-0.326	0.745
Variances				
FRISKY12	9.010	1.277	7.055	0.000
FMON12	0.061	0.011	5.372	0.000
RQUAD	0.000	0.000	999.000	999.000
Residual Variances				
FRISKY13	8.188	0.395	20.711	0.000
FRISKY14	8.188	0.395	20.711	0.000
FRISKY15	8.188	0.395	20.711	0.000
FRISKY16	8.188	0.395	20.711	0.000
FRISKY17	8.188	0.395	20.711	0.000
FRISKY18	8.188	0.395	20.711	0.000
FMON13	0.084	0.004	20.580	0.000
FMON14	0.084	0.004	20.580	0.000
FMON15	0.084	0.004	20.580	0.000
FMON16	0.084	0.004	20.580	0.000
FMON17	0.084	0.004	20.580	0.000
FMON18	0.084	0.004	20.580	0.000
RINT	18.257	2.213	8.249	0.000
RLIN	0.513	0.083	6.180	0.000
MINT	0.191	0.023	8.259	0.000
MLIN	0.010	0.001	7.859	0.000
New/Additional Parameters				
MR2RRSD	**-0.017**	**0.036**	**-0.465**	**0.642**
RR2MRSD	**0.079**	**0.037**	**2.131**	**0.033**

What if we controlled for the concurrent effect of Monitor → Risky before examining the lagged effect of Monitor → Risky (my own preference)?

```
TITLE:   Model 3b: Example of Structured Residuals to Fit M->R Cross-Lagged
Path that controls for concurrent effect before fitting the lagged effect
```

All else is the same until here…

```
! Residual WP effect between same ages, held equal across age
  Frisky12-Frisky18 PON Fmon12-Fmon18 (ResEff);

! Cross-lagged WP effects predicting next occasion, held equal across age
  Frisky13-Frisky18 PON Fmon12-Fmon17 (MR2RR);

MODEL CONSTRAINT:
NEW(ResStD MR2RRsd);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStD = ResEff * SQRT(Mresvar) / SQRT(Rresvar);  ! STD M->R concurrent
  MR2RRsd = MR2RR * SQRT(Mresvar) / SQRT(Rresvar);  ! STD M->R lagged effect
```

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
New/Additional Parameters				
RESSTD	0.339	0.031	10.952	0.000
MR2RRSD	-0.054	0.033	-1.610	0.107

```
TITLE:   Model 3c: Example of Structured Residuals to Fit R->M Cross-Lagged
Path that controls for concurrent effect before fitting the lagged effect
```

All else is the same until here…

```
! Residual WP effect between same ages, held equal across age
  Fmon12-Fmon18 PON Frisky12-Frisky18 (ResEff);

! Cross-lagged WP effects predicting next occasion, held equal across age
  Fmon13-Fmon18 PON Frisky12-Frisky17 (RR2MR);

MODEL CONSTRAINT:
NEW(ResStD RR2MRsd);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  ResStD = ResEff * SQRT(8.3538) / SQRT(0.08077); ! STD R->M concurrent
  RR2MRsd = RR2MR * SQRT(8.3538) / SQRT(0.08077); ! STD R->M lagged effect
```

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
New/Additional Parameters				
RESSTD	0.371	0.031	11.846	0.000
RR2MRSD	0.087	0.034	2.547	0.011

It looks like evidence for a lagged Risky → Monitor lagged effect is a little stronger after controlling for the concurrent effect (and vice-versa).

Here is a comparison of the SEM cross-lagged effects to those using long M-SEM with the LAGGED residual option (available in Mplus 8.1+ with BAYES estimation).

```
TITLE:   Model 3a: Undirected Directed Multivariate Change Model as Long M-SEM
         All fixed L1 effects specified in WITHIN
         Adding lagged effects of both M -> R and R --> M
         using new residual LAGGED option and BAYES estimation

DATA:    FILE = Example5b.csv;   ! Syntax in same folder as data
VARIABLE: ! List of variables in data file
  NAMES = PersonID Att12 occasion age risky mon roundage
          time att4 timesq mon3;
! Variables to be analyzed in this model
  USEVARIABLE = time timesq att4 risky mon3;
  MISSING ARE ALL (-999);   ! Missing data identifier
! MLM options
  CLUSTER = PersonID;             ! Level-2 ID
  BETWEEN = att4;                 ! Observed ONLY level-2 predictors
  WITHIN  = time timesq;          ! Observed ONLY level-1 predictors
  LAGGED  = risky(1) mon3(1);     ! Create Mplus lag-1 variables

ANALYSIS:    TYPE = TWOLEVEL RANDOM; ESTIMATOR = BAYES; BITERATIONS = 50000;
ANALYSIS:    TECH8;  ! Used to examined convergence
MODEL:  ! R = risky behavior, M = monitoring
%WITHIN%
 risky mon3 (Rresvar Mresvar); ! L1 R: Residual variances (labels)
 Rlin  | risky ON time;        ! Placeholder for R linear change
 Rquad | risky ON timesq;      ! Placeholder for R quadratic change
 Mlin  | mon3  ON time;        ! Placeholder for M linear change
 risky WITH mon3   (ResCov);   ! L1 WP covariance for concurrent M->R
 risky^ ON mon3^1  (MRLagEff); ! L1 WP fixed effect of ^residual lagged M->R
 mon3^  ON risky^1 (RMLagEff); ! L1 WP fixed effect of ^residual lagged R->M

%BETWEEN%
[risky mon3 Rlin Rquad Mlin];  ! Fixed intercepts, fixed change slopes
 risky mon3 (Rintvar Mintvar); ! L2 G: Random intercept variances (labels)
 Rlin Mlin  (Rlinvar Mlinvar); ! L2 G: Random linear change variances
 risky Rlin ON att4 (XtoYint XtoYlin);  ! Att-> R int, R linear change
  mon3 Mlin ON att4 (XtoMint XtoMlin);  ! Att-> M int, M linear change
! L2 G: covariances for random intercepts and slopes across outcomes
  risky Rlin mon3 Mlin WITH risky Rlin mon3 Mlin;
 Rquad@0;                      ! No quadratic change variance

MODEL CONSTRAINT:
NEW(MR2RRsd RR2MRsd);
! STD = Unstd * SQRT(Xvar) / SQRT(Yvar)
  MR2RRsd = MRLagEff * SQRT(0.08077) / SQRT(8.3538);  ! STD M->R lagged effect
  RR2MRsd = RMLagEff * SQRT(8.3538)  / SQRT(0.08077); ! STD R->M lagged effect
```

Note that this Model 3a is not fully equivalent to the wide single-level SEM version, which constrained the level-1 residual variances and same-occasion covariance to be equal over time (even though the age 12 versions are unpredicted).

3a SEM via ML and its lagged effects

	Estimate	S.E.	Est./S.E.	Two-Tailed P-Value
New/Additional Parameters				
MR2RRSD	-0.017	0.036	-0.465	0.642
RR2MRSD	0.079	0.037	2.131	0.033 *

3a MLM via Bayes and their residual lagged effects

	Estimate	Posterior S.D.	One-Tailed P-Value	Significance
New/Additional Parameters				
MR2RRSD	-0.017	0.037	0.303	
RR2MRSD	0.081	0.036	0.013	*

For example results sections using this data story and these types of models, please see Hoffman (2015 chapter 9) or Hoffman & Hall (2024)